
Funkcionalna veri�kacija hardvera

Ova veºba je posve¢ena opisu komunikacije izme�u drajvera i sekvencera, opisu sekvenci i se-
quence item-a i opisu raznih mogu¢nosti UVM-a u pogledu pisanja i kori²¢enja sekvenci. Dat je i
opis mehanizma zavr²etka testa.

1 Uvod

U materijalu za veºbu 5 zapo£eli smo pravljenje UVM veri�kacionog okruºenja pomo¢u koga ¢e se
veri�kovati calc digitalni sistem. Prva komponenta koju smo napravili bila je Test klasa i u metodi
main_phase smo postavljali odre�ene vrednosti na ulaze calc digitalnog sistema kako bismo testirali
njegovu funkcionalnost, pri £emu smo pristup ulazima digitalnog sistema dobili preko virtuelnog
interfejsa. Veri�kacija je u potpunosti mogla biti izvr²ena samo u test klasi, ali to ne bi po²tovalo
principe UVM metodologije koji nalaºu particionisanje koda na smislene celine(na vi²e komponenti).
Iz tog razloga u ovom materijalu uvodima dodatne UVM klase: sequence item, sequence, sequencer
driver. Sa ovim promenama veri�kaciono okruºenje bi trebalo da izgleda kao na slici 1:

Slika 1: Veri�kaciono okruºenje, veºba 6

Sa prethodne slike se vidi da Sekvenca (eng. Sequence) i drajver eng. Driver me�usobno komu-
niciraju preko sekvencer komponente (eng. Sequencer), pri £emu se preko sekvencera ²alje objekat
sequence item klase u kome se nalaze informacije neophodne drajveru i sekvenci. U nastavku je
svaka od klasa detaljnije opisana.

4



Funkcionalna veri�kacija hardvera

2 Data (sequence) item

Data (sequence) item predstavlja transakciju koja se koristi kao stimulus. Koristi se na mnogim
mestima u veri�kacionom okruºenju uklju£ujuci slanje podataka od sekvencera ka drajveru. Slan-
jem vi²e ovih transakcija ka drajveru, kreira se stimulus na ulazima DUT-a. Sama transakcija treba
da sadrºi sve informacije potrebne drajveru da ispravno generi²e signale na interfejsu, ali moºe i
sadrºati neke dodatne podatke koji sluºe za kontrolu.

Ove transakcije ¢e se implementirati kao klase koje nasle�uju (direktno ili indirektno) uvm_sequence_item
klasu. Polja u klasi zavise od potreba drajvera, nivoa apstrakcije, ali i potreba ostatka okruºenja.
Uglavnom se ti podaci mogu podeliti u nekoliko grupa:

� Kontrolna polja - npr. tip transfera, veli£ina, ...

� Payload polja - sami podaci koji se ²alju, u slu£aju calc dizajna, to bi bili podaci nad kojima
treba izvr²iti operaciju i tip operacije.

� Kon�guraciona polja - npr. pona²anje prilikom gre²aka

� Polja za analizu - pomo¢na polja koja olak²avaju analizu, npr. vreme

Sam princip generisanja stimulusa u constrained-random veri�kaciji se zasniva na randomizaciji
ovih transackija kako bi se pokrili interesantni scenariji. Samim tim, veliki broj polja je £esto
deklarisan kao rand, uz odre�ena ograni£enja kako bi se generisale ºeljene transakcije.

Sadrºaj transakcija je opisan na primeru transakcije za APB i UART protokol. U ovom primeru
se vidi da transakcija moºe sadrºati ºeljeni broj polja, ali i metode koje olak²avaju njeno kori²¢enje
(u primeru UART seq item-a postoji metoda za ra£unanje parnosti).

`ifndef APB_SEQ_ITEM_SV
`de�ne APB_SEQ_ITEM_SV

parameter ADDR_WIDTH = 32;
parameter RDATA_WIDTH = 32;
parameter WDATA_WIDTH = 32;
typedef enum {

APB_READ = 0,
APB_WRITE = 1
} apb_direction_enum;

class apb_seq_item extends uvm_sequence_item;

// polja
rand bit [ADDR_WIDTH − 1 : 0] addr;
rand apb_direction_enum dir;
rand bit [RDATA_WIDTH − 1 : 0] rdata;
rand bit [WDATA_WIDTH − 1 : 0] wdata;
rand int unsigned delay = 0;
bit error ;

// ogranicenja
constraint c_delay { delay <= 10 ; }

// UVM factory registracija
`uvm_object_utils_begin(apb_seq_item)

`uvm_�eld_int(addr, UVM_DEFAULT)
`uvm_�eld_enum(apb_direction_enum, dir, UVM_DEFAULT)
`uvm_�eld_int(rdata, UVM_DEFAULT)
`uvm_�eld_int(wdata, UVM_DEFAULT)
`uvm_�eld_int(delay, UVM_DEFAULT)
`uvm_�eld_int(error, UVM_DEFAULT)

`uvm_object_utils_end

// konstruktor
function new(string name = "apb_seq_item");

super.new(name);
endfunction : new

endclass

`endif

Kod 1: Primer APB transakcije

`ifndef UART_SEQ_ITEM_SV
`de�ne UART_SEQ_ITEM_SV

typedef enum bit {GOOD_PARITY, BAD_PARITY} parity_e;
parameter PAYLOAD_WIDTH = 8;
parameter STOP_WIDTH = 8;
parameter ERR_WIDTH = 8;

class uart_seq_item extends uvm_sequence_item;

rand bit start_bit;
rand bit [PAYLOAD_WIDTH−1:0] payload;
bit parity ;
rand bit [STOP_WIDTH−1:0] stop_bits;
rand bit [ERR_WIDTH−1:0] error_bits;
rand parity_e parity_type;
rand int unsigned delay;

// ogranicenja
constraint c_delay {delay < 20;}
constraint c_start_bit {start_bit == 0;}
constraint c_stop_bits {stop_bits == '1;}
constraint c_parity_type {parity_type == GOOD_PARITY;}
constraint c_error_bits {error_bits == 0;}

// UVM factory registracija
`uvm_object_utils_begin(uart_seq_item)

`uvm_�eld_int(start_bit, UVM_DEFAULT)
`uvm_�eld_int(payload, UVM_DEFAULT)
`uvm_�eld_int(parity, UVM_DEFAULT)
`uvm_�eld_int(stop_bits, UVM_DEFAULT)
`uvm_�eld_int(error_bits, UVM_DEFAULT)
`uvm_�eld_enum(parity_e,parity_type, UVM_DEFAULT)
`uvm_�eld_int(delay, UVM_DEFAULT)

`uvm_object_utils_end

// konstruktor
function new(string name = "uart_seq_item");

super.new(name);
endfunction

// metoda za racunanje parnosti
function bit calc_parity(int unsigned num_of_data_bits = 8, bit[1:0] parity_mode = 0);

bit parity ;

if (num_of_data_bits == 6)
parity = ^payload[5:0];

else if (num_of_data_bits == 7)
parity = ^payload[6:0];

else
parity = ^payload;

case(parity_mode[0])
0: parity = ~parity;
1: parity = parity;

endcase

case(parity_mode[1])
0: parity = parity;
1: parity = ~parity_mode[0];

endcase

if (parity_type == BAD_PARITY)
return ~parity;

else
return parity ;

endfunction

// parnost se racuna posle randomizacije, na osnovu dodeljenih vrednosti
function void post_randomize();

parity = calc_parity();
endfunction : post_randomize

endclass

`endif

Kod 2: Primer UART transakcije

Nasle�ivanjem iz uvm_sequence_item klase i kori²¢enjem raznih UVM �eld makroa, omogu¢ava
se kori²¢enje mnogobrojnih metoda za rad sa ovom klasom uklju£uju¢i print, copy, compare itd.
Primer ispisa print metode je tako�e dat ispod. Vidi se da kori²¢enje ovih metoda moºe biti veoma
korisno i umnogome olak²ati debug.

# ------------------------------------------------
# Name Type Size Value
# ------------------------------------------------
# uart_frame uart_frame - @366
# start_bit integral 1 'h0
# payload integral 8 'hcc
# parity integral 1 'h1
# stop_bits integral 2 'h3
# error_bit integral 4 'h0
# parity_type parity_e 1 GOOD_PARITY
# delay integral 32 'd13
# ------------------------------------------------

3 Sekvence

Sekvence su objekti (ne komponente) koje sadrºe logiku generisanja stimulusa. Unutar njih se
kreira objekat klase Seqeunce item koji se preko sekvencera prosle�uje drajveru. Kao i ve¢ina
UVM okruºenja, i kod sekvenci postoji preporu£en na£in upotrebe, kako u pogledu strukture same
sekvence, tako i za sam na£in generisanja stimulusa. U ovom poglavlju je opisan mehanizam gener-
isanja stimulusa, dat je pregled strukture sekvenci i obja²njene £esto kori²¢ene funkcije.

3.1 Struktura

Sve sekvence u okruºenju treba da, direktno ili indirektno, nasle�uju uvm_sequence klasu. Po²to
sekvence nisu komponente one ne sadrºe UVM faze, me�utim ipak postoji sli£an mehanizam koji
garantuje ispravan rad i ispravnu komunikaciju sa drajverom. Tri naj£e²¢e kori²¢ene metode su
pre_body(), body() i post_body(). Glavna logika treba da se nalazi u body() tasku. Pre_body() i
post_body() su taskovi koji se pozivaju pre, odnosno posle body() taska. Kao i kod pisanja testova,
dobra je praksa da postoji bazna klasa u kojoj se deklari²e sekvencer, vr²i eventualno podizanje/spu²-
tanje prigovora (raise / drop objection, opisano u narednom poglavlju), kao i sve ostale radnje koje
¢e se ponavljati u svim sekvencama.

class calc_seq extends uvm_sequence#(calc_seq_item);

`uvm_object_utils(calc_seq)
`uvm_declare_p_sequencer(calc_sequencer)

function new(string name = "calc_seq");
super.new(name);

endfunction

virtual task pre_body();
// ...

endtask : pre_body

// transaction generating logic in body
virtual task body();
// calls to uvm_do or uvm_do with macro
// or start / �nish item
// ...

endtask : body

virtual task post_body();
// ...

endtask : post_body

endclass : calc_seq

sekvence je potrebno parametrizovati tipom sequence_item-a koji ¢e se generisati. Ovo, izme�u
ostalog, omogu¢ava kori²¢enje podrazumevanog objekta(req) u sekvencama. Pored factory regis-
tracije (� uvm_object_utils(calc_seq) u primeru), potrebno je i deklarisati sekvencer koji ¢e se
koristiti. Svaka sekvenca se pri pokretanju veºe za odgovaraju¢i sekvencer kome se moºe pristu-
pati preko pokaziva£a. Ovaj korak je mogu¢e i izostaviti, ili deklarisati sekvencere na drugi na£in,
me�utim najjednostavniji na£in za na²e potrebe je da se deklari²e takozvani p sekvencer koriste¢i
odgovaraju¢i makro (� uvm_declare_p_sequencer(calc_sequencer), gde calc_sequencer predstavlja
ime kori²¢enog sekvencera).

3.2 Generisanje stimulusa

Za generisanje transakcije potrebno ispratiti nekoliko koraka:

1. Kreiranje - veoma je bitno kreirati paket koriste¢i factory kako bi se olak²ala ponovna upotreba
odnosno kako bi se dozvolio override metod ukoliko je to potrebno (obja²njeno u narednim
veºbama)

2. Start_item() - blokiraju¢i poziv koji £eka da se uspostavi veza sa drajverom. Sekvenceru ²alje
pokaziva£ na slede¢i objekat

3. Randomizacija paketa (polja unutar sequence item-a) ukoliko je to potrebno

4. Finish_item() - blokiraju¢i poziv koji £eka da drajver zavr²i sa paketom (napomena: start_item()
i �nish_item() £ekaju na drajver kako bi se osigurao korektan transfer paketa, ali oni ne tro²e
simulaciono vreme)

5. Get_response() - opcioni poziv ako je potrebno da drajver po²alje odgovor sekvenceru

Prva £etiri koraka su obavezna, dok je peti opcioni i koristi se samo ukoliko za dato okruºenje
postoji potreba prikupljanja odgovora od drajvera. Logika generisanja stimulusa treba da se nalazi
u body() tasku, npr:

virtual task body();

calc_seq_item calc_it;

// prvi korak kreiranje transakcije
calc_it = calc_seq_item::type_id::create("calc_it");

// drugi korak − start
start_item(calc_it);

// treci korak priprema
// po potrebi moguce prosiriti sa npr. inline ogranicenjima
assert (calc_it .randomize());

// cetvrti korak − �nish
�nish_item(calc_it) ;

endtask: body

Po²to se ova £etiri koraka £esto ponavljaju tj. standardni su za interakciju drajvera i sekvencera,
razvijeni su odgovaraju¢i UVM makroi kako bi se izbeglo nepotrebno ponavljanje koda i pove¢ala
£itljivost. Iako veoma korisni, makroi ipak uvode neka ograni£enja pri kori²¢enju, tako da je nekad
ipak potrebno ru£no odraditi ove korake. �UVM Cookbook� ne preporu£uje upotrebu makroa, ali
su oni £esti u praksi.

3.2.1 � uvm_do

� uvm_do makro kao argument prima sequence_item. Kada se nasledi i parametrizuje uvm_sequence
klasa, nasle�uje se i instanca objekta pod nazivom req £iji je tip upravo parametar koji smo prosledili.
�to zna£i da u svakoj sekvenci ve¢ postoji objekat req koji se moºe koristiti za komunikaciju. Sli£no
postoji i objekat rsp koji treba koristiti za odgovor ukoliko je isti potreban. Upotreba � uvm_do
makroa je:

virtual task body();
`uvm_do(req)

endtask : body

Ovim se postiºe generisanje i slanje jedne transakcije drajveru, koja ¢e se zatim generisati na
interfejsu.

3.2.2 � uvm_do_with

Ukoliko je potrebna ve¢a kontrola prilikom slanja transakcije, mogu¢e je koristiti � uvm_do_with
makro, koji osim sequence_item argumenta prima i bilo koje validno inline ograni£enje. Npr:

virtual task body();
`uvm_do_with(req, { req.data == 16'h5A5A; } )

endtask : body

U ovom primeru se ²alje jedna transakcija pri £emu je vrednost data polja ograni£ena na 16'h5A5A.

3.2.3 Kontrola

U prethodnim primerima smo slali samo jednu transakciju. �esto je neophodno omogu¢iti ve¢u
kontrolu nad izvr²avanjem sekvenci - omogu¢iti kontrolisanu randomizaciju, slanje ºeljenog broja
transakcija itd. Prilikom pisanja sekvenci mogu¢e je koristiti sve mogu¢nosti koje SystemVerilog
pruºa. U nastavku je dato par konstrukcija koje se £esto sre¢u i koje daju uvid u ove mogu¢nosti.

Kontrola prilikom randomizacije je £esto vr²i pomo¢u rand polja u sekvenci, kako bi se omogu¢ila
laka kontrola direktno iz testa koji startuje sekvencu. Slede¢i primer ilustruje slanje vi²e transakcija.

class calc_simple_seq extends calc_base_seq;

`uvm_object_utils (calc_simple_seq)

rand int unsigned num_of_tr;

constraint num_of_tr_con { num_of_tr inside { [1:100] }; }

function new(string name = "calc_simple_seq");
super.new(name);

endfunction

virtual task body();
repeat(num_of_tr) begin

`uvm_do(req);
end

endtask : body

endclass : calc_simple_seq

Tako�e je mogu¢e pozivati jednu sekvencu iz druge, pri £emu je mogu¢e koristiti prethodno
opisane makroe i nad sekvencama (ili koristiti metodu start opisanu u narednom poglavlju).

class calc_simple_seq extends calc_base_seq;

`uvm_object_utils (calc_simple_seq)

calc_sub_seq sub_seq;

function new(string name = "calc_simple_seq");
super.new(name);

endfunction

virtual task body();
`uvm_do(sub_seq);

endtask : body

endclass : calc_simple_seq

3.3 Pokretanje sekvenci

Sekvence se pokre¢u ili iz testa ili iz drugih sekvenci. Postoje dva glavna na£ina pokretanja sekvenci
- eksplicitno koriste¢i start metodu ili implicitno kao podrazumevanu sekvencu.

Ukoliko je potrebna ve¢a kontrola pokretanja sekvenci (npr. pode²avanje parametara, trenutak
startovanja, ...) preporu£uje se eksplicitno pokretanje sekvence kori²¢enjem start metode, npr:

`ifndef TEST_SIMPLE_SV
`de�ne TEST_SIMPLE_SV

class test_simple extends test_base;

`uvm_component_utils(test_simple)

calc_simple_seq simple_seq;

function new(string name = "test_simple", uvm_component parent = null);
super.new(name,parent);

endfunction : new

function void build_phase(uvm_phase phase);
super.build_phase(phase);
simple_seq = calc_simple_seq::type_id::create("simple_seq");

endfunction : build_phase

task main_phase(uvm_phase phase);
phase.raise_objection(this) ;
simple_seq.start(seqr) ;
#100ms;// pre nego sto zavrsimo simulaciju dajemo vremena DUT da izvrsi proracun
phase.drop_objection(this);

endtask : main_phase

endclass

`endif

Kod 3: Start_metoda

Prilikom poziva start metode jedini obavezan argument je sekvencer na kome ¢e pokrenuti
sekvenca. �to se ti£e podrazumevanih sekvenci, one se uglavnom navode u build fazi testa, npr:

`ifndef TEST_SIMPLE_2_SV
`de�ne TEST_SIMPLE_2_SV

class test_simple_2 extends test_base;

`uvm_component_utils(test_simple_2)

function new(string name = "test_simple_2", uvm_component parent = null);
super.new(name,parent);

endfunction : new

function void build_phase(uvm_phase phase);
super.build_phase(phase);

uvm_con�g_db#(uvm_object_wrapper)::set(this,
"seqr.main_phase",
"default_sequence",
calc_simple_seq::type_id::get());

endfunction : build_phase

endclass

`endif

Kod 4: Podrazumevane_sekvence

Za navo�enje podrazumevane sekvence se koriti UVM con�guration database koji omogu¢ava
postavljanje kon�guracionih parametara u okruºenju, u na²em slu£aju podrazumevane sekvence
(uvm_con�g_db ¢e biti detaljnije obra�en u narednim veºbama). U prethodnom primeru je za
main fazu prethodno instanciranog seqr sekvencera pode²ena calc_simple_seq kao podrazumevana
sekvenca.

Pokretanje podrazumevanih sekvenci je u UVM-u nasle�eno iz OVM-a i zadrºano je radi lak²eg
prelaza sa jedne metodologije na drugu. Iako i dalje £esto kori²¢eno, preporuka je koristiti start
metodu radi bolje kontrole.

4 Sekvencer

Sekvencer koristi sekvence kako bi poslao podatke drajveru, ali i prima odgovor od drajvera ukoliko
je to potrebno £ime se omogu¢ava kreiranje korisnih stimulusa. Sekvencer sluºi kao arbiter £iji je
zadatak kontrola slanja transakcija iz jedne ili vi²e sekvenci.

U praksi naj£e²¢e nije potrebno modi�kovati sekvencer jer su funkcionalnosti koje se nalaze u
uvm_sequencer klasi obi£no dovoljne. Kao i drajver, i sekvencer je mogu¢e parametrizovati sa tipom
transakcije, npr:

class calc_sequencer extends uvm_sequencer#(calc_seq_item);
// ...

endclass : calc_sequencer

Ukoliko, osim parametrizacije, nije potrebno ubacivati dodatni sadrºaj u sekvencer, £esto se samo
koristi typedef naredba kako bi se izbegao nepotrebni kod, npr:

typedef uvm_sequencer#(calc_seq_item) calc_sequencer;

5 Drajver

Drajver je aktivna komponenta koja emulira signale koji se ²alju dizajnu. Na osnovu podataka
iz sequence item-a i semplovanja signala na interfejsu, postavlja signale na ulaze dizajna koji se
veri�kuje. Transakcije koje prima su obi£no na vi²em nivou apstrakcije, pa se vr²i konverzija na
pin-nivo apstrakcije kako bi se ispravno generisali signali. Drajver je preko TLM porta povezan na
sekvencer kako bi vr²io komunikaciju i sadrºi virtuelni interfejs kako bi generisao signale. U UVM-u
se drajver implementira nasle�uju¢i uvm_driver #(REQ,RSP) klasu, koja je parametrizovana tipom
transakcija koju ¢e drajver zahtevati i odgovarati (engl. request, response). Ukoliko se navede samo
REQ i RSP ¢e biti istog tipa kao i REQ. Npr:

class calc_driver extends uvm_driver#(calc_seq_item);
// ...

endclass : calc_driver

Napomena: u SystemVerilog-u se parametrizacija vr²i koriste¢i �#� znak. Nalik C++-su ovim
mehanizmom se omogu¢ava pisanje generi£kih klasi kojima ¢e se stvarna vrednost parametra proslediti
tek prilikom kori²¢enja. Primer deklarisanja parametrizovane klase:

class param_example #(type T = int, type G = bit, int A = 10);

Kao i sve komponente unutar UVM veri�kacionog okruºenja, i drajver se implementira tako ²to
se nasledi uvm_driver klasa, parametrizovana tipom sequence item-a koji ¢e se koristiti za zahtev
odnosno odgovor. Pored ovog drajver treba da sadrºi kod za factory registraciju, konstruktor, kao i
virtuelni interfejs preko koga ¢e generisati signale. Kostur ve¢ine drajvera je dat ispod.

class example_driver extends uvm_driver#(example_seq_item);

`uvm_component_utils(example_driver)

virtual interface example_if vif;

function new(string name = "example_driver", uvm_component parent = null);
super.new(name,parent);

endfunction

function void connect_phase(uvm_phase phase);
super.connect_phase(phase);
if (!uvm_con�g_db#(virtual example_if)::get(this, "*", "example_if", vif))
`uvm_fatal("NO_IF",{"virtual interface must be set for: ",get_full_name(),".vif"})

endfunction : connect_phase

task main_phase(uvm_phase phase);
forever begin
seq_item_port.get_next_item(req);
drive_tr();
seq_item_port.item_done();

end
endtask : main_phase

task drive_tr();
// do actual driving here
// ...

endtask : drive_tr

endclass : example_driver

Nasle�ivanjem uvm_driver klase, nasle�uje se i TLM port za komunikaciju sa sekvencerom
(seq_item_port), ali i req i rsp objekti koji se mogu koristiti za generisanje stimulusa. Ovi objekti
¢e poprimiti tip kojim je parametrizovan drajver (example_seq_item u primeru).

5.1 API

API za komunikaciju sa sekvencerom je:

� get_next_item - blokiraju¢a metoda koja £eka da REQ transakcija postane dostupna i vra¢a
pokaziva£ na taj objekat; nakon izvr²avanja ove metode obavezno je pozvati item_done pre
slede¢eg poziva get_next_item

� try_next_item - neblokiraju¢a metoda koja proverava da li postoji transakcija. Vra¢a null
ukoliko nema dostupne transakcije u sekvenceru

� item_done - neblokiraju¢a metoda koja �nalizira handshake sa sekvencerom; treba je pozvati
nakon get_next_item ili uspe²nog try_next_item; mogu¢e joj je proslediti RSP transakciju
kao argument

� peek - blokiraju¢a metoda koja £eka na dostupnu REQ transakciju od sekvencera i vra¢a pokazi-
va£ na taj objekat; naredni pozivi ka peek pre get ili item_done poziva ¢e vratiti pokaziva£ na
isti REQ objekat

� get - blokiraju¢a metoda koja £eka na dostupnu REQ transakciju od sekvencera; kada primi
transakciju kompletira handshake i vra¢a pokaziva£ na REQ objekat

� put - neblokiraju¢a metoda koja vra¢a RSP transakciju kao odgovor sekvenceru; moºe se
pozvati u bilo kom trenutku i nije deo handshake-a izme�u drajvera i sekvencera

Na osnovu ovih metoda, postoje dva na£ina da se ostvari komunikacija sa sekvencerom - koriste¢i
get_next_item / item_done ili get / put mehanizam.

Ispravno generisanje stimulusa u UVM-u zavisi od dobre veze izme�u sekvenci i drajvera. Sekvence
i drajver se moraju usaglasiti kako ne bi do²lo do deadlock situacija kada se jedna strana zauvek
blokira £ekaju¢i na odgovor. Kako bi se izbegle ovakve situacije i omogu¢ila laka upotreba, ako i
ponovna upotreba drajvera potrebno je dobro izmodelovati i dokumentovati funkcionalnost drajvera.
Postoji veliki broj modela za generisanje stimulusa, me�utim neki od naj£e²¢e kori²¢enjih modela
za drajvere su:

� Unidirectional Non-Pipelined - zahtevi se ²alju drajveru, ali se ne primaju odgovori od drajvera;
drajver moºe koristiti neki vid rukovanja, ali je proces slanja podataka uvek u jednom smeru;
ovaj model se koristi za jednostrane komunikacione kanale nalik PCM interfejsu

� Bidirectional Non-Pipelined - podaci se ²alju u oba smera, sekvencer ²alje transakcije drajveru,
dok drajver ²alje odgovor; uvek je samo jedna transakcija aktivna, odgovor se uvek ²alje pre
slanja slede¢e transakcije; ovaj model se koristi za jednostavnije protokole npr. APB

� Pipelined - podaci se ²alju u oba smera, ali se faza slanja zahteva preklapa sa fazom slanja
odgovora na prethodni zahtev; proto£na obrada omogu¢ava bolje performanse, ali komplikuje
sam kod jer se zahtevi i odgovori moraju posebno obra�ivati; primer upotrebe bio bi za AHB
protokol

� Out Of Order Pipelined - u nekim primenama odgovori se ne ²alju u redosledu u kom su
primeljni zahtevi; ovo rezultuje u dosta komplikovanijem modelu gde se uglavnom koriste
redovi za pra¢enje odgovora; jo² jedna varijacija bila bi burst transferi za vi²e transakcija
istovremeno; primer upotebe npr. AXI

U nastavku je su data dva primera modela drajvera. Za kompletan opis svih modela kao i alter-
nativne implementacije datih modela pogledati �UVM Cookbook�, poglavlje �Driver/Use Models�.

5.1.1 Unidirectional Non-Pipelined

Kod ovog modela drajver kontroli²e �ow koriste¢i get_next_item poziv kako bi primio slede¢u
transakciju, a poziv item_done metode se vr²i tek kada je zavr²eno sa procesiranjem transakcije.

U nastavku je dat kori²¢enja ovog modela i odgovaraju¢e sekvence. U primeru se ²alju ADPCM
paketi koriste¢i PCM protokol.

Slika 2: Protokol

class unidir_driver extends uvm_driver #(unidir_seq_item);

`uvm_component_utils(unidir_driver)

virtual unidir_if vif ;

function new(string name = "unidir_driver", uvm_component parent = null);
super.new(name, parent);

endfunction

task main_phase(uvm_phase phase);
int top_idx = 0;
// pocetno stanje
vif .frame <= 0;
vif .data <= 0;

forever begin
seq_item_port.get_next_item(req); // zahtev za transakcijom
repeat(req.delay) begin // kasnjenje izmedju paketa

@(posedge vif.clk) ;
end

vif .frame <= 1; // pocetak slanja
for( int i = 0; i < 8; i++) begin // slanje podataka

@(posedge vif.clk) ;
vif .data <= req.data[3:0];
req.data = req.data >> 4;

end

vif .frame <= 0; // kraj slanja
seq_item_port.item_done(); // transakcija je zavrsena

end
endtask: run

endclass : unidir_driver

// generisanje transakcija u petlji
class unidir_tx_seq extends uvm_sequence #(unidir_seq_item);

`uvm_object_utils(unidir_tx_seq)

// unidir sequence_item
unidir_seq_item req;

// kontrola za broj transakcija
rand int no_reqs = 10;

function new(string name = "unidir_tx_seq");
super.new(name);

endfunction

task body();
req = unidir_seq_item::type_id::create("req");
repeat(no_reqs) begin

start_item(req);
assert (req.randomize());
�nish_item(req);

end
endtask: body

endclass: unidir_tx_seq

Kod 5: Primer use modela

5.1.2 Bidirectional Non-Pipelined

Naj£e²¢e kori²¢en model je upravo bidirekcioni. Sekvencer ²alje zahteve drajveru koji ih izvr²ava, a
zatim ²alje odgovor nazad. Nova transakcija ne moºe biti zapo£eta do god se ne primi odgovor i ne
zavr²i prethodna. Ovi modeli se koriste za jednostavnije protokole kao ²to je npr. AMPA APB.

Drajver prima transakciju, po protokolu koji se implementira po²alje podatke uz eventualno
£ekanje i zatim postavi polja u transakciji koja sluºe za odgovor i zavr²i rukovanje sa sekvencerom.
Po²to se i u drajveru i u sekvenci koristi pokaziva£ na isti objekat, nakon ²to se sekvenca odblokira
(posle �nish_item) moºe koristiti objekat sa novim vrednostima i vr²iti dalju analizu. Primer je dat
ispod:

Slika 3: Protokol

class bidir_driver extends uvm_driver #(bidir_seq_item);

`uvm_component_utils(bidir_driver)

virtual bidir_if vif ;

function new(string name = "bidir_driver", uvm_component parent = null);
super.new(name, parent);

endfunction

task main_phase(uvm_phase phase);
// pocetno stanje
vif . valid <= 0;

vif .rnw <= 1;
// cekanje da se reset zavrsi
@(posedge vif.resetn) ;

forever begin
seq_item_port.get_next_item(req); // zahtev za transakcijom

repeat(req.delay) begin
@(posedge vif.clk) ;

end
vif . valid <= 1;
vif .addr <= req.addr;
vif .rnw <= req.read_not_write;
if (req.read_not_write == 0) begin

vif .write_data <= req.write_data;
end
while(vif .ready != 1) begin

@(posedge vif.clk) ;
end

// na kraju transakcije podesiti polja za odgovor
if (req.read_not_write == 1) begin

req.read_data = vif.read_data; // vratiti procitani podatak, ukoliko je u pitanju citanje
end
req. error = vif . error ; // podesiti status
vif . valid <= 0; // kraj slanja

seq_item_port.item_done(); // transakcija je zavrsena
end

endtask

endclass

class bidir_seq extends uvm_sequence #(bidir_seq_item);

`uvm_object_utils(bidir_seq)

bidir_seq_item req;

rand int limit = 40; // broj iteracija

function new(string name = "bidir_seq");
super.new(name);

endfunction

task body();
req = bidir_seq_item::type_id::create("req");

repeat(limit ) begin
start_item(req);
// adresa je ogranicena na validan opseg
assert (req.randomize() with {addr inside {[16'h0100:16'h1000]};});
�nish_item(req);
// req pokazuje na objekat sa novim vrednostima podesenim u drajveru
uvm_report_info("seq_body", req.convert2string());

end
endtask

endclass

Kod 6: Primer use modela

6 Komunikacija

UVM komponente komuniciraju preko standardnih TLM (engl. transaction level modeling) interfejsa
koji olak²ava povezivanje i ponovno kori²¢enje. Komponenta moºe komunicirati preko svog inter-
fejsa sa bilo kojom drugom komponentom koja implementira taj interfejs. U praksi je ovaj na£in
komunikacije veoma £est za drajver - sekvencer interakciju i monitor - scoreboard komunikaciju. U
ovoj i narednoj veºbi je fokus na komunikaciji izme�u drajvera i sekvencera, dok je detaljan opis
rada TLM konekcija ostavljen za veºbu o monitoru.

U uvm_driver i uvm_sequencer klasama postoji deklaracija odgovaraju¢ih portova i ugra�ena
logika oko komunikacije, tako da je upotreba od strane korisnika veoma jednostavna. Potrebno je
samo povezati drajver i sekvencer, koriste¢i metodu connect. Povezivanje se tipi£no vr²i u connect
fazi, gde je, na primeru ispod, driver instanca prethodno kreiranog drajvera, a sequencer instanca
prethodno kreiranog sekvencera. seq_item_port i seq_item_export su imena nasle�enih TLM por-
tova.

function void connect_phase(uvm_phase phase);
drviver .seq_item_port.connect(sequencer.seq_item_export);

endfunction : connect_phase

Sam UVM ne pruºa samo mogu¢nost kori²¢enja bibilioteke, ve¢ i de�ni²e na£in upotrebe kako bi
se kreirale univerzalne veri�kacione komponente koje ¢e pratiti datu metodologiju. S tim u vezi, pos-
toji jasno de�nisana komunikacija izme�u drajvera i sekvencera, koriste¢i odre�ene UVM funkcije,
£iji je opis dat u nastavku.

UVM drajver sadrºi mnoge metode za komunikaciju sa sekvencerom. Naj£e²¢e kori²¢ene su:

� get_next_item() - blokiraju¢a metoda koja traºi novi objekat; vra¢a pokaziva£ na novi REQ
objekat

� try_next_item() - neblokiraju¢a verzija koja vra¢a null pokaziva£ ukoliko nema dostupnog
objekta

� item_done() - neblokiraju¢a metoda koja zavr²ava proces rukovanja izme�u drajvera i sekvencera;
poziva se nakon get_next_item() ili uspe²nog try_next_item()

Sam sekvencer sluºi za arbitraciju, dok se sequence item objekti kreiraju i kontroli²u iz sekvenci.
Da bi se paket poslao drajveru, on mora pro¢i kroz nekoliko koraka:

1. Kreiranje - veoma je bitno kreirati paket koriste¢i factory kako bi se olak²ala ponovna upotreba
odnosno kako bi se dozvolio override metod ukoliko je to potrebno (obja²njeno u narednim
veºbama)

2. Start_item() - blokiraju¢i poziv koji £eka da se uspostavi veza sa drajverom. Sekvenceru ²alje
pokaziva£ na slede¢i objekat

3. Randomizacija paketa ukoliko je to potrebno

4. Finish_item() - blokiraju¢i poziv koji £eka da drajver zavr²i sa paketom (napomena: start_item()
i �nish_item() £ekaju na drajver kako bi se osigurao korektan transfer paketa, ali oni ne tro²e
simulaciono vreme)

5. Get_response() - opcioni poziv ako je potrebno da drajver po²alje odgovor sekvenceru

Preporu£eni �ow je prikazan na slici 4. Drajver traºi transakciju od sekvencera koriste¢i get_next_item(),
obra�uje istu i zatim kompletira handshake koriste¢i item_done() £ime signalizira sekvenceru da je
objekat obra�en. Sa druge strane, u sekvenci se kreiraju transakcije prate¢i prethodne korake. Kada
drajver pozove item_done(), odblokira¢e se �nish_item() metoda u sekvenci i moºe se nastaviti sa

Slika 4: Drajver-sekvencer �ow

izvr²avanjem.

Pored ovog na£ina, mogu¢e je ostvariti komunikaciju koriste¢i peek(), put() i get() metode u
drajveru koje su opisane u narednoj veºbi.

7 Mehanizam zavr²etka testa

Svaki testben£ koji koristi standardne UVM faze se sastoji od nekoliko faza za kreiranje i povezivanje
koje se izvr²avaju u nultom vremenu, zatim odre�en broj faza koje tro²e vreme, kao i nekoliko faza
za zavr²etak testa koje se izvr²avaju u nultom vremenu. Do kraja testa ¢e do¢i kada se zavr²e sve
faze koje tro²e vreme, a svaka faza se zavr²ava kada nema prigovora za tu fazu ²to zna£i da se kraj
testa kontroli²e podizanjem i spu²tanjem prigovora. Komponenta ili sekvenca ¢e podi¢i prigovor na
po£etku aktivnosti koja se mora zavr²iti u datoj fazi, a spustiti prigovor po zavr²etku te aktivnosti.
Tek kada se svi podignuti prigovori spuste, mogu¢e je zavr²iti datu fazu. U praksi se £esto de²ava
da je faza samo trenutno bez prigovora tj. da ¢e se uskoro podi¢i jo² neki prigovor i da ne treba
odmah zavr²iti sa trenutnom fazom. U tom slu£aju se moºe postaviti vreme £ekanja (engl. drain
time) koje unosi ka²njenje za zavr²etak faze. Ako u tom vremenskom intervalu podigne prigovor,
faza se ne¢e zavr²iti i opet ¢e se £ekati na spu²tanje svih prigovora.

Kontrola prigovora se uglavnom vr²i ili iz samog testa ili iz sekvenci. Kontrola iz testa se £esto
sre¢e kada se sekvence eksplicitno pokre¢u. Jedan primer je prikazan ispod.

task main_phase(uvm_phase phase);
phase.raise_objection(); //rasing objection
example_seq.start(example_agent.sequencer);
phase.drop_objection(); //droping objection

endtask

�to se ti£e kontrole iz sekvenci, po²to se £itava logika generisanja stimulusa nalazi u body()
tasku, prigovor je potrebno podi¢i pre odnosno posle izvr²avanja body() taska i ovo se naj£e²¢e radi
u pre_body() odnosno post_body() taskovima. Ovaj kod se naj£e²¢e nalazi u baznoj sekvenci kako bi
se prigovori generisali prilikom pokretanja svake sekvence. Test se tada ne¢e zavr²iti do god postoje
sekvence u okruºenju koje nisu zavr²ile sa generisanjem stimulusa.

// objections are raised in pre_body
virtual task pre_body();
uvm_phase phase = get_starting_phase();
if (phase != null)
phase.raise_objection(this , {"Running sequence '", get_full_name(), "'"});
uvm_test_done.set_drain_time(this, 200ns)

endtask : pre_body

// objections are dropped in post_body
virtual task post_body();
uvm_phase phase = get_starting_phase();
if (phase != null)
phase.drop_objection(this, {"Completed sequence '", get_full_name(), "'"});

endtask : post_body

Jedan na£in podizanja / spu²tanja prigovora je prikazan kodom iznad. Koriste se raise_objection/drop_objection
metode. starting_phase je £lan bazne klase koji ¢e biti ispravno postavljen jedino ukoliko je sekvenca
pokrenuta kao podrazumevana (default) za neku fazu.

Pomo¢u metode set_drain_time se postavlja takozvano drain vreme koje u datom primeru unosi
toleranciju od 200 ns, odnosno nakon spu²tanja poslednjeg prigovora test se ne¢e odmah zavr²iti
nego ¢e se sa£ekati jo² 200 ns zbog mogu¢nosti da se u tom periodu ponovo podigne neki prigovor.

Napomena: po²to je sekvenca ru£no pokrenuta, starting_phase polje iz prethodnog kodnog frag-
menta bi imalo vrednost null tako da se kori²¢enjem oba kodna fragmenta ispovremeno ne bi podigla
dva prigovora za istu sekvencu.

Napomena: podizanje / spu²tanje prigovora u sekvenci se £esto koristi uz UVM 1.1 biblioteku.
Za UVM 1.2 preporu£en na£in je podizanje / spu²tanje prigovora iz testa.

8 Organizacija fajlova koji £ine veri�kaciono okruºenje

Broj fajlova koji £ine veri�kaciono okruºenje moºe da postane dosta veliki i stoga odre�eni alati(kao
Vivado) mogu da imaju problem da shvate kojim redosledom treba da kompajliraju fajlove. Iz tog
razloga alatu treba malo pomo¢i.
U svakom direktorijumu koji £ini veri�kaciono okruºenje preporuka je da postoji Systemverilog
paket koji uklju£uje(eng. include) sve fajlove iz tog direktorijuma (sem top fajla), i koji tako�e
uklju£uje(eng. import) fajlove iz drugih direktorijuma ukoliko ih koriste fajlovi iz trenutnog di-
rektorijuma. Fajlove treba uvla£iti po²tuju¢i hijerarhiju veri�kacionog okruºenja, fajlovi najniºi u
hijerarhiji se navode prvi(ukoliko se uvu£e test pre sekvence, kompajliranje ¢e pu¢i jer test instancira
sekvencu koja nije kompajlirana).

U priloºenom materijalu za ove veºbe nalazi se verif direktorijum sa svim fajlovima koji £ine
skelet veri�kacionog okruºenja za veri�kaciju calc dizajna:

Slika 5: Hijerarhija direktorijuma, veºba 6

Taj direktorijum unutar sebe poseduje calc_test_pkg.sv fajl unutar koga su uklju£eni svi fajlovi
iz verif direktorijuma koji £ine veri�kaciono okruºenje, ²to ilustruje slede¢i kodni listing:

`ifndef CALC_TEST_PKG_SV
`de�ne CALC_TEST_PKG_SV

package calc_test_pkg;

import uvm_pkg::*; // import the UVM library
`include "uvm_macros.svh" // Include the UVM macros

import calc_agent_pkg::*;
import calc_seq_pkg::*;

`include "v6_test_base.sv"
`include "v6_test_simple.sv"
`include "v6_test_simple_2.sv"

endpackage : calc_test_pkg

`include "calc_if .sv"

`endif

Kod 7: v6_calc_test_pkg

Svaki pod-direktorijum unutar verif direkorijuma tako�e poseduje Systemverilog paket koji un-
utar sebe uklju£uje sve fajlove iz tog direktorijuma. Iz prethodnog kodnog listinga se moºe videti
da calc_test_pkg pored fajlova iz datog direktorijuma uvla£i fajlove iz drugih direktorijuma:

import calc_agent_pkg::*;
import calc_seq_pkg::*;

Razlog za ovo je taj ²to unutar klase test instanciramo objekte klasa driver, sequence i sequencer
koji su de�nisani u drugim direktorijumima. Naredba import calc_agent_pkg::* importuje sve
fajlove koji su uvu£eni unutar calc_agent_pkg.sv fajla. Naravno, mogli smo da uvu£emo i samo
odre�enu klasu iz pomenutog pod-direktorijuma na slede¢i na£in:

import calc_agent_pkg::calc_driver;

Obratite paºnju na redosled uvla£enja fajlova u prethodnom kodnom listingu. Po²to klasa
test_base unutar sebe instancira drajver i sekvencer, pre nego ²to se uvu£e fajl u kome je de�nisana
test base klasa moraju se importovati fajlovi u kojima su de�nisani drajver i sekvencer. Ukoliko bi
se uradilo slede¢e, kompajliranje ne bi pro²lo:

`ifndef CALC_TEST_PKG_SV
`de�ne CALC_TEST_PKG_SV

package calc_test_pkg;

import uvm_pkg::*; // import the UVM library
`include "uvm_macros.svh" // Include the UVM macros

`include "test_base.sv" // greska, moraju prvo da se importuju fajlovi iz agent paketa i sequence paketa !!!!
import calc_agent_pkg::*;
import calc_seq_pkg::*;

`include "test_simple.sv"
`include "test_simple_2.sv"

endpackage : calc_test_pkg

`include "calc_if .sv"

`endif

Kod 8: v6_calc_test_pkg sa gre²kom

Napomena: U test_pkg.sv fajlu moºe se primetiti da je calc_if.sv fajl uvu£en nakon de�nicije
paketa. To je ura�eno jer Systemverilog ima ograni£enje po kome interface ne sme biti de�nisan
unutar paketa.

8.1 Kompajliranje u Vivadu

Ukoliko se koristi Vivado za kompajliranje fajlova, prilikom kreiranja projekta potrebno je da se
ubace dizajn fajlovi, paketi u kojima su uklju£eni svi fajlovi koji £ine veri�kacioni okruºenje i veri-
�kacioni top fajl. Na slede¢oj slici prikazan je primer dodavanja dizajn fajlova koji £ine calc digitalni
sistem kao i veri�kacionih fajlova koji se koriste za veri�kaciju:

Crvenom strelicom je nazna£eno da svi dizajn fajlovi treba da se koriste za sintezu i simulaciju,
dok se fajlovi koji £ine veri�kaciono okruºenje koriste samo pri simulaciji.

Slika 6: Vivado, dodavanje fajlova pri kreiranju projekta

8.2 Kompajliranje u Cadence Xcelium simulatoru

Ukoliko se za simulaciju koristi Xcelium simulator, onda je prilikom pisanja .f fajla neophodno
uklju£iti samo dizajn fajlove, pakete i top fajl. Slede¢i primer ilustruje .f fajl pomo¢u koga se
kompajliraju fajlovi iz priloºenog materijala za veºbu 6:

−uvmhome "/eda/cadence/2019−20/RHELx86/XCELIUM_19.03.013/tools/methodology/UVM/CDNS−1.2/"
−uvm +UVM_TESTNAME=test_simple
−sv +incdir+./verif
−sv +incdir+./verif/Agent
−sv +incdir+./verif/Sequences

./dut/alu_input_stage.v

./dut/alu_output_stage.v

./dut/exdbin_mac.v

./dut/holdreg.v

./dut/mux_out.v

./dut/shifter .v

./dut/priority .v

./dut/calc_top.v

−sv ./verif/Agent/calc_agent_pkg.sv
−sv ./verif/Sequences/calc_seq_pkg.sv
−sv ./verif/calc_test_pkg.sv

−sv ./verif/calc_verif_top.sv

#−LINEDEBUG
−access +rwc

−disable_sem2009
−nowarn "MEMODR"
−timescale 1ns/10ps

Kod 9: v6_run.f

9 Zadaci

Zadatak Napisati sekvence koje ¢e se koristiti za veri�kaciju �Calc1� DUT-a, uklju£uju¢i:

� Sekvenca za generisanje jedne transakcije na nasumi£no izabranom portu, sa nasumi£nim po-
dacima i komandom

� Sekvenca za generisanje vi²e nasumi£nih transakcija na istom portu

� Sekvenca za generisanje vi²e nasumi£nih transakcija, pri £emu se broj transakcija nalazi u
opsegu (1, 10), a svaka naredna transakcija se nalazi na portu razli£itom od prethodnog

� Sekvenca za generisanje vi²e nasumi£nih transakcija pri £emu su komande uvek za jednu ALU,
bez ka²njenja izme�u generisanja transakcija

� smisliti jo² nekoliko primera sekvenci koji bi mogle biti korisne za veri�kaciju

Zadatak Napisati test (uz eventualne pomo¢ne testove ili sekvence) u kome ¢e se poslati ta£no 8
transakcija, koriste¢i neku od prethodno razvijenih sekvenci. Na koje je na£ine ovo mogu¢e odraditi?

U sekciji Appendix i u priloºenom materijalu nalazi se skelet veri�kacionog okruºenja za proveru
ispravnosti rada calc dizajna. Pro²iriti veri�kaciono okruºenje tako da se provere prethodno navedeni
scenariji.

10 Appendix

`ifndef CALC_IF_SV
`de�ne CALC_IF_SV

interface calc_if (input clk , logic [6 : 0] rst ) ;

parameter DATA_WIDTH = 32;
parameter RESP_WIDTH = 2;
parameter CMD_WIDTH = 4;

logic [DATA_WIDTH − 1 : 0] out_data1;
logic [DATA_WIDTH − 1 : 0] out_data2;
logic [DATA_WIDTH − 1 : 0] out_data3;
logic [DATA_WIDTH − 1 : 0] out_data4;
logic [RESP_WIDTH − 1 : 0] out_resp1;
logic [RESP_WIDTH − 1 : 0] out_resp2;
logic [RESP_WIDTH − 1 : 0] out_resp3;
logic [RESP_WIDTH − 1 : 0] out_resp4;
logic [CMD_WIDTH − 1 : 0] req1_cmd_in;
logic [DATA_WIDTH − 1 : 0] req1_data_in;
logic [CMD_WIDTH − 1 : 0] req2_cmd_in;
logic [DATA_WIDTH − 1 : 0] req2_data_in;
logic [CMD_WIDTH − 1 : 0] req3_cmd_in;
logic [DATA_WIDTH − 1 : 0] req3_data_in;
logic [CMD_WIDTH − 1 : 0] req4_cmd_in;
logic [DATA_WIDTH − 1 : 0] req4_data_in;

endinterface : calc_if

`endif

Kod 10: calc_if

`ifndef CALC_SEQUENCER_SV
`de�ne CALC_SEQUENCER_SV

class calc_sequencer extends uvm_sequencer#(calc_seq_item);

`uvm_component_utils(calc_sequencer)

function new(string name = "calc_sequencer", uvm_component parent = null);
super.new(name,parent);

endfunction

endclass : calc_sequencer

`endif

Kod 11: v6_calc_sequencer

`ifndef CALC_DRIVER_SV
`de�ne CALC_DRIVER_SV
class calc_driver extends uvm_driver#(calc_seq_item);

`uvm_component_utils(calc_driver)
virtual interface calc_if vif ;
function new(string name = "calc_driver", uvm_component parent = null);

super.new(name,parent);
endfunction // new

function void build_phase(uvm_phase phase);
if (!uvm_con�g_db#(virtual calc_if)::get(null, "*", "calc_if", vif ))
`uvm_fatal("NOVIF",{"virtual interface must be set:",get_full_name(),".vif"})

endfunction // build_phase

task main_phase(uvm_phase phase);
forever begin

@(posedge vif.clk) ;
if (! vif . rst )
begin

seq_item_port.get_next_item(req);

`uvm_info(get_type_name(),
$sformatf("Driver sending ...\ n%s", req.sprint ()) ,
UVM_HIGH)
vif .req1_data_in = req.operand1;
vif .req1_cmd_in = req.cmd;
@(posedge vif.clk) ;
vif .req1_data_in = req.operand2;
vif .req1_cmd_in = 0;

seq_item_port.item_done();
end
end

endtask : main_phase

endclass : calc_driver

`endif

Kod 12: v6_calc_driver

`ifndef CALC_SEQ_ITEM_SV
`de�ne CALC_SEQ_ITEM_SV

class calc_seq_item extends uvm_sequence_item;

/* TODO add �elds and methods here */

`uvm_object_utils_begin(calc_seq_item)
`uvm_object_utils_end

function new(string name = "calc_seq_item");
super.new(name);

endfunction

endclass : calc_seq_item

`endif

Kod 13: v6_calc_seq_item

`ifndef CALC_AGENT_PKG
`de�ne CALC_AGENT_PKG

package calc_agent_pkg;

import uvm_pkg::*;
`include "uvm_macros.svh"

//////////////////////////////////////////////////////////
// include Agent components : driver,monitor,sequencer
/////////////////////////////////////////////////////////
`include "v6_calc_seq_item.sv"
`include "v6_calc_sequencer.sv"
`include "v6_calc_driver.sv"

endpackage

`endif

Kod 14: v6_calc_agent_pkg

`ifndef TEST_BASE_SV
`de�ne TEST_BASE_SV

class test_base extends uvm_test;

`uvm_component_utils(test_base)

calc_driver drv;
calc_sequencer seqr;

function new(string name = "test_base", uvm_component parent = null);
super.new(name,parent);

endfunction : new

function void build_phase(uvm_phase phase);
super.build_phase(phase);
drv = calc_driver::type_id::create("drv", this ) ;
seqr = calc_sequencer::type_id::create("seqr", this ) ;

endfunction : build_phase

function void connect_phase(uvm_phase phase);
drv.seq_item_port.connect(seqr.seq_item_export);

endfunction : connect_phase

endclass : test_base

`endif

Kod 15: v6_test_base

`ifndef TEST_SIMPLE_SV
`de�ne TEST_SIMPLE_SV

class test_simple extends test_base;

`uvm_component_utils(test_simple)

calc_simple_seq simple_seq;

function new(string name = "test_simple", uvm_component parent = null);
super.new(name,parent);

endfunction : new

function void build_phase(uvm_phase phase);
super.build_phase(phase);
simple_seq = calc_simple_seq::type_id::create("simple_seq");

endfunction : build_phase

task main_phase(uvm_phase phase);
phase.raise_objection(this) ;
simple_seq.start(seqr) ;
#100ms;// pre nego sto zavrsimo simulaciju dajemo vremena DUT da izvrsi proracun
phase.drop_objection(this);

endtask : main_phase

endclass

`endif

Kod 16: v6_test_simple

`ifndef TEST_SIMPLE_2_SV

`de�ne TEST_SIMPLE_2_SV

class test_simple_2 extends test_base;

`uvm_component_utils(test_simple_2)

function new(string name = "test_simple_2", uvm_component parent = null);
super.new(name,parent);

endfunction : new

function void build_phase(uvm_phase phase);
super.build_phase(phase);

uvm_con�g_db#(uvm_object_wrapper)::set(this,
"seqr.main_phase",
"default_sequence",
calc_simple_seq::type_id::get());

endfunction : build_phase

endclass

`endif

Kod 17: v6_test_simple_2

`ifndef CALC_TEST_PKG_SV
`de�ne CALC_TEST_PKG_SV

package calc_test_pkg;

import uvm_pkg::*; // import the UVM library
`include "uvm_macros.svh" // Include the UVM macros

import calc_agent_pkg::*;
import calc_seq_pkg::*;

`include "v6_test_base.sv"
`include "v6_test_simple.sv"
`include "v6_test_simple_2.sv"

endpackage : calc_test_pkg

`include "calc_if .sv"

`endif

Kod 18: v6_calc_test_pkg

`ifndef CALC_BASE_SEQ_SV
`de�ne CALC_BASE_SEQ_SV

class calc_base_seq extends uvm_sequence#(calc_seq_item);

`uvm_object_utils(calc_base_seq)
`uvm_declare_p_sequencer(calc_sequencer)

function new(string name = "calc_base_seq");
super.new(name);

endfunction

// objections are raised in pre_body
virtual task pre_body();

uvm_phase phase = get_starting_phase();
if (phase != null)
phase.raise_objection(this , {"Running sequence '", get_full_name(), "'"});

endtask : pre_body

// objections are dropped in post_body
virtual task post_body();

uvm_phase phase = get_starting_phase();
if (phase != null)
phase.drop_objection(this, {"Completed sequence '", get_full_name(), "'"});

endtask : post_body

endclass : calc_base_seq

`endif

Kod 19: v6_calc_base_seq

`ifndef CALC_SIMPLE_SEQ_SV
`de�ne CALC_SIMPLE_SEQ_SV

class calc_simple_seq extends calc_base_seq;

`uvm_object_utils (calc_simple_seq)

function new(string name = "calc_simple_seq");
super.new(name);

endfunction

virtual task body();
// simple example − just send one item
`uvm_do(req);

endtask : body

endclass : calc_simple_seq

`endif

Kod 20: v6_calc_simple_seq

`ifndef CALC_SEQ_PKG_SV
`de�ne CALC_SEQ_PKG_SV
package calc_seq_pkg;

import uvm_pkg::*; // import the UVM library
`include "uvm_macros.svh" // Include the UVM macros
import calc_agent_pkg::calc_seq_item;
import calc_agent_pkg::calc_sequencer;

`include "v6_calc_base_seq.sv"
`include "v6_calc_simple_seq.sv"

endpackage
`endif

Kod 21: v6_calc_seq_pkg

`ifndef CALC_TEST_PKG_SV
`de�ne CALC_TEST_PKG_SV

package calc_test_pkg;

import uvm_pkg::*; // import the UVM library
`include "uvm_macros.svh" // Include the UVM macros

import calc_agent_pkg::*;
import calc_seq_pkg::*;

`include "test_base.sv"
`include "test_simple.sv"
`include "test_simple_2.sv"

endpackage : calc_test_pkg

`include "calc_if .sv"

`endif

Kod 22: v6_calc_test_pkg

module calc_verif_top;

import uvm_pkg::*; // import the UVM library
`include "uvm_macros.svh" // Include the UVM macros

import calc_test_pkg::*;

logic clk ;
logic [6 : 0] rst ;

// interface
calc_if calc_vif(clk , rst ) ;

// DUT
calc_top DUT(

.c_clk ( clk ) ,

. reset ( rst ) ,

.out_data1 ( calc_vif .out_data1 ),

.out_data2 ( calc_vif .out_data2 ),

.out_data3 ( calc_vif .out_data3 ),

.out_data4 ( calc_vif .out_data4 ),

.out_resp1 ( calc_vif .out_resp1 ),

.out_resp2 ( calc_vif .out_resp2 ),

.out_resp3 ( calc_vif .out_resp3 ),

.out_resp4 ( calc_vif .out_resp4 ),

.req1_cmd_in ( calc_vif.req1_cmd_in ),

.req1_data_in ( calc_vif.req1_data_in ),

.req2_cmd_in ( calc_vif.req2_cmd_in ),

.req2_data_in ( calc_vif.req2_data_in ),

.req3_cmd_in ( calc_vif.req3_cmd_in ),

.req3_data_in ( calc_vif.req3_data_in ),

.req4_cmd_in ( calc_vif.req4_cmd_in ),

.req4_data_in ( calc_vif.req4_data_in )
) ;

// run test
initial begin
uvm_con�g_db#(virtual calc_if)::set(null, "*", "calc_if", calc_vif) ;
run_test();

end

// clock and reset init .
initial begin
clk <= 0;
rst <= 1;
#50 rst <= 0;

end

// clock generation
always #50 clk = ~clk;

endmodule : calc_verif_top

Kod 23: v6_calc_verif_top

